

# SAWSCRIPT

Galois, Inc. 421 SW 6th Ave., Ste. 300 Portland, OR 97204

### Abstract

We introduce the SAWScript language, aiming to provide a programmable interface to Galois's formal verification technologies, covering the Cryptol, Java, and LLVM languages. We use various simple programs as examples, taken from the domain of cryptography. Our proofs use SAWScript to equivalence check functions written in one language against their counterparts in another, or reference implementations in one language against more efficient production implementations in the same language.

© 2014 Galois, Inc.



## Contents

Introduction	3
Example: Find First Set	3
Reference Implementation	3
Optimized Implementation	3
Buggy Implementation	4
Generating LLVM Code	4
Equivalence Proof	4
Cross-Language Proofs	6
Using SMT-Lib Solvers	7
External SAT Solvers	8
Compositional Proofs	9
Other Examples	10
Java Equivalence Checking	10
AIG Export and Import	10



#### Introduction

SAWScript is a special-purpose programming language developed by Galois to help orchestrate and track the results of the large collection of proof tools necessary for analysis and verification of complex software artifacts.

The language adopts the functional paradigm, and largely follows the structure of many other typed functional languages, with some special features specifically targeted at the coordination of verification and analysis tasks.

This tutorial introduces the details of the language by walking through several examples, and showing how simple verification tasks can be described.

## **Example: Find First Set**

As a first example, we consider a simple function that identifies the first 1 bit in a word. The function takes an integer as input, treated as a vector of bits, and returns another integer which indicates the index of the first bit set. This function exists in a number of standard C libraries, and can be implemented in several ways.

#### Reference Implementation

One simple implementation take the form of a loop in which the index starts out at zero, and we keep track of a mask initialized to have the least significant bit set. On each iteration, we increment the index, and shift the mask to the left. Then we can use a bitwise "and" operation to test the bit at the index indicated by the index variable. The following C code (which is also in the code/ffs.c file accompanying this tutorial) uses this approach.

```
uint32_t ffs_ref(uint32_t word) {
    int i = 0;
    if(!word)
        return 0;
    for(int cnt = 0; cnt < 32; cnt++)
        if(((1 << i++) & word) != 0)
        return i;
    return 0; // notreached
}</pre>
```

This implementation is relatively straightforward, and a proficient C programmer would probably have little difficulty believing its correctness. However, the number of branches taken during execution could be as many as 32, depending on the input value. It's possible to implement the same algorithm with significantly fewer branches, and no backward branches.

#### **Optimized Implementation**

An alternative implementation, taken by the following program (also in code/ffs.c), treats the bits of the input word in chunks, allowing sequences of zero bits to be skipped over more quickly.

```
uint32_t ffs_imp(uint32_t i) {
   char n = 1;
   if (!(i & 0xffff)) { n += 16; i >>= 16; }
   if (!(i & 0x00ff)) { n += 8; i >>= 8; }
   if (!(i & 0x000f)) { n += 4; i >>= 4; }
   if (!(i & 0x0003)) { n += 2; i >>= 2; }
```



```
return (i) ? (n+((i+1) & 0x01)) : 0;
}
```

However, this code is much less obvious than the previous implementation. If it is correct, we would like to use it, since it has the potential to be faster. But how do we gain confidence that it calculates the same results as the original program?

SAWScript allows us to state this problem concisely, and to quickly and automatically prove the equivalence of these two functions for all possible inputs.

#### **Buggy Implementation**

Finally, a buggy implementation which is correct on all but one possible input (also in code/ffs.c). Although contrived, this program represents a case where traditional testing – as opposed to verification – is unlikely to be helpful.

```
uint32_t ffs_bug(uint32_t word) {
    // injected bug:
    if(word == 0x101010) return 4; // instead of 5
    return ffs_ref(word);
}
```

SAWScript allows us to quickly identify an input exhibiting the bug.

#### **Generating LLVM Code**

The SAWScript interpreter knows how to analyze LLVM code, but most programs are originally written in a higher-level language such as C, as in our example. Therefore, the C code must be translated to LLVM, using something like the following command:

```
# clang -c -emit-llvm -o ffs.bc ffs.c
```

This command, and following command examples in this tutorial, can be run from the code directory accompanying the tutorial document. A Makefile also exists in that directory, providing quick shortcuts for tasks like this. For instance, we can get the same effect as the previous command by doing:

```
# make ffs.bc
```

#### **Equivalence Proof**

We now show how to use SAWScript to prove the equivalence of the reference and implementation versions of the FFS algorithm, and exhibit the bug in the buggy implementation.

A SAWScript program is typically structured as a set of commands within a main function, potentially along with other functions defined to abstract over commonly-used combinations of commands.

The following script (in code/ffs\_llvm.saw) is sufficient to automatically prove the equivalence of the ffs\_ref and ffs\_imp functions, and identify the bug in ffs\_bug.

```
let main = do {
  print "Extracting reference term";
l <- llvm_load_module "ffs.bc";</pre>
```

```
ffs_ref <- llvm_extract l "ffs_ref" llvm_pure;</pre>
print "Extracting implementation term";
ffs_imp <- llvm_extract l "ffs_imp" llvm_pure;</pre>
print "Extracting buggy term";
ffs_bug <- llvm_extract l "ffs_bug" llvm_pure;</pre>
print "Proving equivalence";
let thm1 = \{\{ \x -> ffs\_ref x == ffs\_imp x \}\};
result <- prove abc thm1;
print result;
print "Finding bug via sat search";
let thm2 = {{ \xspace x -> ffs_ref x != ffs_bug x }};
result <- sat abc thm2;
print result;
print "Finding bug via failed proof";
let thm3 = \{\{ \x -> ffs_ref x == ffs_bug x \}\};
result <- prove abc thm3;
print result;
print "Done.";
```

In this script, the print commands simply display text for the user. The <code>llvm\_extract</code> command instructs the SAWScript interpreter to perform symbolic simulation of the given C function (e.g., <code>ffs\_ref</code>) from a given bitcode file (e.g., <code>ffs\_bc</code>), and return a term representing the semantics of the function. The final argument, <code>llvm\_pure</code> indicates that the function to analyze is a "pure" function, which computes a scalar return value entirely as a function of its scalar parameters.

The let statement then constructs a new term corresponding to the assertion of equality between two existing terms. The prove\_print command can verify the validity of such an assertion, and print out the results of verification. The abc parameter indicates what theorem prover to use.

If the saw executable is in your PATH, you can run the script above with

```
# saw ffs_llvm.saw
```

producing the output

```
Loading module Cryptol
Loading file "ffs_llvm.saw"
Extracting reference term
Extracting implementation term
Extracting buggy term
Proving equivalence
Valid
Finding bug via sat search
Sat: 1052688
Finding bug via failed proof
Invalid: 1052688
Done.
```



Note that 0x101010 = 1052688, and so both explicitly searching for an input exhibiting the bug (with sat) and attempting to prove the false equivalence (with prove) exhibit the bug. Symmetrically, we could use sat to prove the equivalence of ffs\_ref and ffs\_imp, by checking that the corresponding disequality is unsatisfiable. Indeed, this exactly what happens behind the scenes: prove abc <goal>):

#### **Cross-Language Proofs**

We can implement the FFS algorithm in Java with code almost identical to the C version.

The reference version (in code/FFS.java) uses a loop, like the C version:

```
static int ffs_ref(int word) {
   int i = 0;
   if(word == 0)
       return 0;
   for(int cnt = 0; cnt < 32; cnt++)
       if(((1 << i++) & word) != 0)
       return i;
   return 0;
}</pre>
```

And the efficient implementation uses a fixed sequence of masking and shifting operations:

```
static int ffs_imp(int i) {
    byte n = 1;
    if ((i & 0xffff) == 0) { n += 16; i >>= 16; }
    if ((i & 0x00ff) == 0) { n += 8; i >>= 8; }
    if ((i & 0x000f) == 0) { n += 4; i >>= 4; }
    if ((i & 0x0003) == 0) { n += 2; i >>= 2; }
    if (i != 0) { return (n+((i+1) & 0x01)); } else { return 0; }
}
```

Although in this case we can look at the C and Java code and see that they perform almost identical operations, the low-level operators available in C and Java do differ somewhat. Therefore, it would be nice to be able to gain confidence that they do, indeed, perform the same operation.

We can do this with a process very similar to that used to compare the reference and implementation versions of the algorithm in a single language.

First, we compile the Java code to a JVM class file.

```
# javac -g FFS.java
```

Now we can do the proof both within and across languages (from code/ffs\_compare.saw):

```
import "ffs.cry";
let main = do {
    j <- java_load_class "FFS";
    java_ffs_ref <- java_extract j "ffs_ref" java_pure;
    java_ffs_imp <- java_extract j "ffs_imp" java_pure;

l <- llvm_load_module "ffs.bc";
    c_ffs_ref <- llvm_extract l "ffs_ref" llvm_pure;</pre>
```

```
c_ffs_imp <- llvm_extract l "ffs_imp" llvm_pure;</pre>
    print "java ref <-> java imp";
    let thm1 = \{\{ \x -> java_ffs_ref x == java_ffs_imp x \}\};
    prove_print abc thm1;
    print "c ref <-> c imp";
    let thm2 = \{\{ \x -> c_ffs_ref x == c_ffs_imp x \}\};
    prove_print abc thm2;
    print "java imp <-> c imp";
    let thm3 = \{\{ \x -> java_ffs_imp x == c_ffs_imp x \}\};
    prove_print abc thm3;
    print "cryptol imp <-> c imp";
    let thm4 = \{\{ \x -> ffs_imp x == c_ffs_imp x \}\};
    prove_print abc thm4;
    print "cryptol imp <-> cryptol ref";
    let thm5 = \{\{ \x -> ffs_ip x == ffs_ref x \}\};
    prove_print abc thm5;
    print "Done.";
};
```

We can run this with the -j flag to tell it where to find the Java standard libraries:

```
# saw -j <path to rt.jar or classes.jar from JDK> ffs_compare.saw
```

## **Using SMT-Lib Solvers**

The examples presented so far have used the internal proof system provided by SAWScript, based primarily on a version of the ABC tool from UC Berkeley linked into the saw executable. However, there is internal support for other proof tools – such as Yices and CVC4 as illustrated in the next example – and more general support for exporting models representing theorems as goals in the SMT-Lib language. These exported goals can then be solved using an external SMT solver.

Consider the following C file:

```
int double_ref(int x) {
    return x * 2;
}
int double_imp(int x) {
    return x << 1;
}</pre>
```

In this trivial example, an integer can be doubled either using multiplication or shifting. The following SAWScript program (code/double.saw) verifies that the two are equivalent using both internal ABC, Yices, and CVC4 modes, and by exporting an SMT-Lib theorem to be checked later, by an external SAT solver.

```
double_ref <- llvm_extract l "double_ref" llvm_pure;
let thm = {{ \x -> double_ref x == double_imp x }};

r <- prove abc thm;
print r;

r <- prove yices thm;
print r;

r <- prove cvc4 thm;
print r;

let thm_neg = {{ \x -> ~(thm x) }};
write_smtlib1 "double.smt" thm_neg;
print "Done.";
};
```

The new primitives introduced here are the tilde operator, ~, which constructs the logical negation of a term, and write\_smtlib1, which writes a term as a proof obligation in SMT-Lib version 1 format. Because SMT solvers are satisfiability solvers, negating the input term allows us to interpret a result of "unsatisfiable" from the solver as an indication of the validity of the term. The prove primitive does this automatically, but for flexibility the write\_smtlib1 primitive passes the given term through unchanged, because it might be used for either satisfiability or validity checking.

The SMT-Lib export capabilities in SAWScript are currently based on a somewhat outdated implementation, and don't support the full range of operations that the abc tactic support. This will improve in the near future. (The internal support for Z3, CVC4, MathSAT, and Yices is implemented via the Data.SBV package, separately from the internal ABC support via Data.AIG. The two implementations have a similar high-level structure . . . )

#### **External SAT Solvers**

In addition to the abc, cvc4, and yices proof tactics used above, SAWScript can also invoke arbitrary external SAT solvers that support the DIMACS CNF format for problem and solution descriptions, using the external\_cnf\_solver tactic. For example, you can use PicoSAT to prove the theorem thm from the last example, with the following commands:

```
let picosat = external_cnf_solver "picosat" ["%f"];
prove_print picosat thm;
```

The use of let is simply a convenient abbreviation. The following would be equivalent:

```
prove_print (external_cnf_solver "picosat" ["%f"]) thm;
```

The first argument to external\_cnf\_solver is the name of the executable. It can be a fully-qualified name, or simply the bare executable name if it's in your PATH. The second argument is an array of command-line arguments to the solver. Any occurrence of %f is replaced with the name of the temporary file containing the CNF representation of the term you're proving.



The external\_cnf\_solver tactic is based on the same underlying infrastructure as the abc tactic, and is generally capable of proving the same variety of theorems.

## **Compositional Proofs**

The examples shown so far treat programs as monolithic entities. A Java method or C function, along with all of its callees, is translated into a single mathematical model. SAWScript also has support for more compositional proofs, as well as proofs about functions that use heap data structures.

As a simple example of compositional reasoning, consider the following Java code.

```
class Add {
   public int add(int x, int y) {
      return x + y;
   }

   public int dbl(int x) {
      return add(x, x);
   }
}
```

Here, the add function computes the sum of its arguments. The dbl function then calls add to double its argument. While it would be easy to prove that dbl doubles its argument by following the call to add, it's also possible in SAWScript to prove something about add first, and then use the results of that proof in the proof of dbl, as in the following SAWScript code (code/java\_add.saw).

```
let setup : JavaSetup () = do {
    x <- java_var "x" java_int;
    y <- java_var "y" java_int;
    java_return {{ x + y }};
    java_verify_tactic abc;
};
let setup' : JavaSetup () = do {
    x <- java_var "x" java_int;
    java_return {{ x + x }};
    java_verify_tactic abc;
};
let main = do {
    cls <- java_load_class "Add";</pre>
    ms <- java_verify cls "add" [] setup;</pre>
    ms' <- java_verify cls "dbl" [ms] setup';</pre>
    print "Done.";
};
```

This can be run as follows:

```
# saw -j <path to rt.jar or classes.jar from JDK> java_add.saw
```

In this example, the definitions of setup and setup' provide extra information about how to configure the symbolic simulator when analyzing Java code. In this case, the setup blocks provide explicit descriptions of the implicit configuration used by java\_extract (used in the earlier Java FFS example and in the next section).



The java\_var commands instruct the simulator to create fresh symbolic inputs to correspond to the Java variables x and y. Then, the java\_return commands indicate the expected return value of the each method, in terms of existing models (which can be written inline).

Finally, the java\_verify\_tactic command indicates what method to use to prove that the Java methods do indeed return the expected value. In this case, we use ABC.

To make use of these setup blocks, the <code>java\_verify</code> command analyzes the method corresponding to the class and method name provided, using the setup block passed in as a parameter. It then returns an object that describes the proof it has just performed. This object can be passed into later instances of <code>java\_verify</code> to indicate that calls to the analyzed method do not need to be followed, and the previous proof about that method can be used instead of re-analyzing it.

## Other Examples

The code directory contains a few additional examples not mentioned so far. These remaining examples don't cover significant new material, but help fill in some extra use cases that are similar, but not identical to those already covered.

#### Java Equivalence Checking

The previous examples showed comparison between two different LLVM implementations, and cross-language comparisons between Cryptol, Java, and LLVM. The script in code/ffs\_java.saw compares two different Java implementations, instead.

```
let main = do {
  print "Extracting reference term";
  j <- java_load_class "FFS";
  ffs_ref <- java_extract j "ffs_ref" java_pure;

  print "Extracting implementation term";
  ffs_imp <- java_extract j "ffs_imp" java_pure;

  print "Proving equivalence";
  let thm1 = {{ \x -> ffs_ref x == ffs_imp x }};
  prove_print abc thm1;
  print "Done.";
};
```

As with previous Java examples, this one needs to be run with the -j flag to tell the interpreter where to find the Java standard libraries.

```
# saw -j <path to rt.jar or classes.jar from JDK> ffs_java.saw
```

#### **AIG Export and Import**

Most of the previous examples have used the abc tactic to discharge theorems. This tactic works by translating the given term to And-Inverter Graph (AIG) format, transforming the graph in various ways, and then using a SAT solver to complete the proof.

Alternatively, the write\_aig command can be used to write an AIG directly to a file, for later processing by external tools, as shown in code/ffs\_gen\_aig.saw.



```
let main = do {
    cls <- java_load_class "FFS";
    bc <- llvm_load_module "ffs.bc";
    java_ffs_ref <- java_extract cls "ffs_ref" java_pure;
    java_ffs_imp <- java_extract cls "ffs_imp" java_pure;
    c_ffs_ref <- llvm_extract bc "ffs_ref" llvm_pure;
    c_ffs_imp <- llvm_extract bc "ffs_imp" llvm_pure;
    write_aig "java_ffs_ref.aig" java_ffs_ref;
    write_aig "java_ffs_imp.aig" java_ffs_imp;
    write_aig "c_ffs_ref.aig" c_ffs_ref;
    write_aig "c_ffs_ref.aig" c_ffs_imp;

    print "Done.";
};</pre>
```

Conversely, the read\_aig command can construct an internal term from an existing AIG file, as shown in code/ffs\_compare\_aig.saw.

```
let main = do {
    java_ffs_ref <- read_aig "java_ffs_ref.aig";
    java_ffs_imp <- read_aig "java_ffs_imp.aig";
    c_ffs_ref <- read_aig "c_ffs_ref.aig";
    c_ffs_imp <- read_aig "c_ffs_imp.aig";

let thm1 = {{ \x -> java_ffs_ref x == java_ffs_imp x }};
    prove_print abc thm1;

let thm2 = {{ \x -> c_ffs_ref x == c_ffs_imp x }};
    prove_print abc thm2;

print "Done.";
};
```

Both of these scripts can be run by saw with no arguments (though the latter must be run second, because it uses files generated by the former):

```
# saw ffs_gen_aig.saw
# saw ffs_compare_aig.saw
```